

### 3.4 Modular Structural Operational Semantics (MSOS)

Modular structural operational semantics (MSOS) was introduced, as its name implies, to address the non-modularity aspects of (big-step and/or small-step) SOS. There are several reasons why big-step and small-step SOS are non-modular, as well as several facets of non-modularity in general. These are discussed in detail in Section 3.9. In short, a definitional framework is *non-modular* when, in order to add a new feature to an existing language or calculus, one needs to revisit and change some or all of the already defined unrelated features. For example, suppose that one wants to add an output statement to IMP, as we do in IMP++ in Section 3.8. Then one needs to add a new semantic component in the IMP configurations, both in the big-step and in the small-step SOS definitions, to hold the output “buffer”. That means, in particular, that *all* the existing big-step and/or small-step SOS rules of IMP need to change, which is, of course, very inconvenient.

Before we get into the technicalities of MSOS, one natural question to address is why we need modularity of language definitions. One may argue that defining a programming language is a major endeavor, done once and for all, so having to go through the defining rules many times is, after all, not such a bad idea, because it gives one the chance to find and fix potential errors in them. Here are several reasons why modularity is desirable in language definitions:

- Having to modify many or all rules whenever a new rule is added that modifies the structure of the configuration is actually more error prone than it may seem, because rules become heavier to read and debug; for example, one can write  $\sigma$  instead of  $\sigma'$  in a right-hand-side of a rule and a different or wrong language is defined.
- When designing a new language, as opposed to an existing well-understood language, one needs to experiment with features and combinations of features; having to do lots of unrelated changes whenever a new feature is added to or removed from the language burdens the language designer with boring tasks taking considerable time that could have been otherwise spent on actual interesting language design issues.
- There is a plethora of domain-specific languages these days, generated by the need to abstract away from low-level programming language details to important, domain-specific aspects of the application. Therefore, there is a need for rapid language design and experimentation for various domains. Moreover, domain-specific languages tend to be very dynamic, being added or removed features frequently as the domain knowledge evolves. It would be very nice to have the possibility to “drag-and-drop” language features in one’s language, such as functions, exceptions, objects, etc.; however, in order for that to be possible, modularity is crucial.

Whether MSOS gives us such a desired and general framework for modular language design is and will probably always be open for debate. However, at our knowledge MSOS is the first framework that explicitly recognizes the importance of modular language design and provides explicit support to achieve it in the context of SOS. Reduction semantics with evaluation contexts (see Section 3.8.1) was actually proposed before MSOS and also offers modularity in language semantic definitions, but its modularity comes as a consequence of a different way to propagate reductions through language constructs and not as an explicit goal that it strives to achieve.

There are both big-step and small-step variants of MSOS, but we discuss only small-step MSOS. We actually generically call MSOS the small-step, implicitly-modular variant of MSOS (see Section 3.4.3). To bring modularity to SOS, the MSOS proposes the following:

- Separate the syntax (i.e., the fragment of program under consideration) from the non-syntactic components in configurations, and treat them differently, as explained below;
- Make the transitions relate syntax to syntax only (as opposed to configurations), and “hide” the non-syntactic components in a transition label, as explained below;
- Encode in the transition label all the changes in the non-syntactic components of the configuration that need to be applied together with the syntactic reduction given by the transition;
- Use specialized notation in transition labels together with a discipline to refer to the various semantic components and to say that some of them stay unchanged; also, labels can be explicitly or implicitly shared by the conditions and the conclusion of a rule, elegantly capturing the idea that “changes are propagated” through desired language constructs.

A transition in MSOS is of the form

$$P \xrightarrow{\Delta} P'$$

where  $P$  and  $P'$  are programs or fragments of programs and  $\Delta$  is a *label describing the semantic configuration components both before and after the transition*. Specifically,  $\Delta$  is a record containing fields denoting the semantic components of the configuration. The preferred notation in MSOS for stating that in label  $\Delta$  the semantic component associated to the field name *field* *before* the transition takes place is  $\alpha$  is “ $\Delta = \{\text{field} = \alpha, \dots\}$ ”. Similarly, the preferred notation for stating that the semantic component associated to field *field* *after* the transition takes place is  $\alpha$  is “ $\Delta = \{\text{field}' = \alpha, \dots\}$ ” (the field name is primed). For example, the second MSOS rule for variable assignment (when the assigned arithmetic expression is already evaluated) is (this is rule (MSOS-ASGN) in Figure 3.18):

$$x := i \xrightarrow{\{\text{state}=\sigma, \text{state}'=\sigma[i/x], \dots\}} \text{skip}$$

It is easy to “desugar” the rule above into a more familiar SOS rule of the form:

$$\langle x := i, \sigma \rangle \rightarrow \langle \text{skip}, \sigma[i/x] \rangle$$

The above is precisely the rule (SMALLSTEP-ASGN) in the small-step SOS of IMP (see Figure 3.12). The MSOS rule is actually more modular than the SOS one, because of the “...”, which say that “everything else stays unchanged” in the configuration. For example, if we want to extend the language with an output statement as we do within IMP++ in Section 3.8, then a new component, the “output buffer”, needs to be added to the configuration to collect and store the output. Then the SOS rule above needs to be changed into a rule of the form

$$\langle x := i, \sigma, \omega \rangle \rightarrow \langle \text{skip}, \sigma[i/x], \omega \rangle$$

where  $\omega$  is the output buffer, which stays unchanged during the variable assignment operation, while the MSOS rule does not need to be touched.

To impose a better discipline on the use of labels, at the same time making the notation even more compact, MSOS splits the fields into three categories: *read-only*, *read-write* and *write-only*. The field **state** above was write-only, meaning that the transition label can both read its value before the transition takes place and write its value after the transition takes place. Unlike the state, which needs to be both read and written, there are semantic configuration components that only need to be written, as well as ones that only need to be read. In these cases, it is recommended to use

write-only or read-only fields. Read-only fields are only inspected by the rule, but not modified, so they only appear unprimed in labels. For example, when reading the location of a variable in an environment, the environment is not modified. We do not discuss read-only fields further here. The write-only fields are used to record data that is not analyzable during the execution of the program, such as the output or the trace. Their names are always primed and they have a free monoid semantics – everything written on them is actually added to the end. Consider, for example, the extension of IMP with output statements in Section 3.8, whose MSOS second rule (after the argument is evaluated, namely rule (MSOS-OUTPUT) in Section 3.8) is:

$$\text{output}(i) \xrightarrow{\{\text{output}'=i, \dots\}} \text{skip}$$

Compare the rule above with the one below, which uses a read-write attribute instead:

$$\text{output}(i) \xrightarrow{\{\text{output}=\omega, \text{output}'=\omega i, \dots\}} \text{skip}$$

Indeed, mentioning the  $\omega$  is unnecessary, error-prone (e.g., one may forget to add it to the primed field, or may write  $i\omega$  instead of  $\omega i$ ), and non-modular (e.g., one may want to change the binary construct of the output monoid, say to write  $\omega \cdot i$  instead of  $\omega i$ , etc.).

MSOS achieves modularity in two ways: 1) by making intensive use of the record comprehension notation “...”, which, as discussed, indicates that more fields could follow but that they are not of interest for this transition, in particular that they stay unchanged after the application of the transition; and 2) by reusing the same label or portion of label both in the condition and the conclusion of a rule (in particular, if record comprehension is used in both the condition and the conclusion of an MSOS rule, then all the occurrences of “...” stand for the same portion of label, that is, the same fields bound to the same semantic components. For example, the following MSOS rules for first-statement reduction in sequential composition are equivalent and say that all the configuration changes generated by reducing  $s_1$  to  $s'_1$  are propagated when reducing  $s_1 ; s_2$  to  $s'_1 ; s_2$ :

$$\begin{array}{c} \frac{s_1 \xrightarrow{\Delta} s'_1}{s_1 ; s_2 \xrightarrow{\Delta} s'_1 ; s_2} \\[1em] \frac{s_1 \xrightarrow{\{\dots\}} s'_1}{s_1 ; s_2 \xrightarrow{\{\dots\}} s'_1 ; s_2} \\[1em] \frac{s_1 \xrightarrow{\{\text{state}=\sigma, \dots\}} s'_1}{s_1 ; s_2 \xrightarrow{\{\text{state}=\sigma, \dots\}} s'_1 ; s_2} \end{array}$$

Indeed, advancing one step the first statement in a sequential composition of statements has the same effects on the configuration as if the statement was advanced the same one step in isolation, without the other statement involved; said differently, the side effects are all properly propagated.

MSOS (the implicitly-modular variant of it, see Section 3.4.3) has been refined to actually allow for dropping such redundant labels like above from rules. In other words, if a label is missing from a transition then the “implicit label” is assumed: if the rule is unconditional then the implicit label

is the identity label (in which the primed fields have the values as the corresponding unprimed ones, etc.), while if the rule is conditional, then the condition and the conclusion transitions share the same label, that is, they perform the same changes on the semantic components of the configuration. With this new notational convention, the most elegant way to write the rule above in MSOS is:

$$\frac{s_1 \rightarrow s'_1}{s_1 ; s_2 \rightarrow s'_1 ; s_2}$$

This is precisely the rule (MSOS-SEQ-ARG1) in Figure 3.18, part of the MSOS semantics of IMP.

One of the important merits of MSOS is that it captures formally many of the “tricks” that language designers informally use to avoid writing awkward and heavy SOS definitions.

Additional notational shortcuts are welcome in MSOS if properly explained and made locally rigorous, without having to rely on other rules. For example, the author of MSOS finds the rule

$$x \xrightarrow{\{\text{state}, \dots\}} \text{state}(x)$$

to be an acceptable variant of the lookup rule:

$$x \xrightarrow{\{\text{state}=\sigma, \dots\}} \sigma(x)$$

despite the fact that, strictly speaking,  $\text{state}(x)$  does not make sense by itself (recall that **state** is a field name, not the state function) and that field names are expected to be paired with their semantic component in labels. Nevertheless, there is only one way to make sense of this rule, namely to replace any use of **state** by its semantic contents, which therefore does not need to be mentioned.

A major goal when using MSOS to define languages or calculi is to write on the labels as little information as possible and to use the implicit conventions for the missing information. That is because everything written on labels may work against modularity if the language is later on extended or simplified. As an extreme case, if one uses only read/write fields in labels and mentions all the fields together with all their semantic contents on every single label, then MSOS becomes conventional SOS and therefore suffers from the same limitations as SOS in what regards modularity.

Recall the rules in Figure 3.13 for deriving the transitive closure  $\rightarrow^*$  of the small-step SOS relation  $\rightarrow$ . In order for two consecutive transitions to “compose”, the source configuration of the second had to be identical to the target configuration of the first. A similar property must also hold in MSOS, otherwise one may derive inconsistent computations. This process is explained in MSOS by making use of category theory (Section 2.10), associating MSOS labels with morphisms in a special category and then using the morphism composition mechanism of category theory.

However, category theory is not needed in order to understand how MSOS works in practice. A simple way to explain its label composition is by translating, or “desugaring” MSOS definitions into SOS, as we implicitly implied when we discussed the MSOS rule for variable assignment above. Indeed, once one knows all the fields in the labels, which happens once a language definition is complete, one can automatically associate a standard small-step SOS definition to the MSOS one by replacing each MSOS rule with an SOS rule over configurations containing, besides the syntactic contents, the complete semantic contents extracted from the notational conventions in the label. The resulting SOS configurations will not have fields anymore, but will nevertheless contain all the semantic information encoded by them. For example, in the context of a language containing only a state and an output buffer as semantic components in its configuration, the four rules discussed

above for variable assignment, output, sequential composition and lookup desugar into the following conventional SOS rules, as expected (listed in the discussed order):

$$\begin{aligned} \langle x := i, \sigma, \omega \rangle &\rightarrow \langle \text{skip}, \sigma[i/x], \omega \rangle \\ \langle \text{output}(i), \sigma, \omega \rangle &\rightarrow \langle \text{skip}, \sigma, \omega i \rangle \\ \frac{\langle s_1, \sigma, \omega \rangle \rightarrow \langle s'_1, \sigma', \omega' \rangle}{\langle s_1 ; s_2, \sigma, \omega \rangle \rightarrow \langle s'_1 ; s_2, \sigma', \omega' \rangle} \\ \langle x, \sigma, \omega \rangle &\rightarrow \langle \sigma(x), \sigma, \omega \rangle \end{aligned}$$

Note that for unconditional MSOS rules the meaning of the missing label fields is “stay unchanged”, while in the case of conditional rules the meaning of the missing fields is “same changes in conclusion as in the condition”.

This way of regarding MSOS as a convenient “front-end” to SOS also supports the introduction of further notational conventions in MSOS if desired, like the one using `state(x)` instead of  $\sigma(x)$  in the right-hand side of the transition above, as far as one clearly explains how such conventions are desugared when going from MSOS to SOS. Finally, the translation of MSOS into SOS also allows MSOS to “borrow” from SOS the reflexive/transitive closure  $\rightarrow^*$  of the one-step relation.

### 3.4.1 The MSOS of IMP

Figures 3.17 and 3.18 show the MSOS definition of IMP. There is not much to comment on the MSOS rules in these figures, except, perhaps, to note how compact and elegant they are compared to the corresponding SOS definition in Figures 3.11 and 3.12. Except for the three rules (MSOS-LOOKUP), (MSOS-ASGN) and (MSOS-VARS), which make use of labels, they are as compact as they can be in any SOS-like setting for any language including the defined constructs. Also, the above-mentioned three rules only mention those components from the labels that they really need, so they allow for possible extensions of the language, as we will do with IMP++ in Section 3.8.

Since the MSOS proof system in Figures 3.17 and 3.18 translates, following the informal procedure described above, in the SOS proof system in Figures 3.11 and 3.12, basically all the small-step SOS intuitions and discussions for IMP in Section 3.3.2 carry over here almost unchanged. In particular:

**Definition 18.** *We say that  $C \rightarrow C'$  is derivable with the MSOS proof system in Figures 3.17 and 3.18, written  $\text{MSOS}(\text{IMP}) \vdash C \rightarrow C'$ , iff  $\text{SMALLSTEP}(\text{IMP}) \vdash C \rightarrow C'$  (using the proof system in Figures 3.11 and 3.12). Similarly,  $\text{MSOS}(\text{IMP}) \vdash C \rightarrow^* C'$  iff  $\text{SMALLSTEP}(\text{IMP}) \vdash C \rightarrow^* C'$ .*

**Exercise 48.** *Change the MSOS rules for `/` so that it short-circuits when  $a_1$  evaluates to 0.*

**Exercise 49.** *Change the MSOS rules of the IMP conjunction so that it is not short-circuited.*

**Exercise 50.** *Give an alternative MSOS definition of `while` which wastes no computational step.*

**Exercise 51.** *Add `error` expressions and statements and modify the MSOS of IMP in Figures 3.17 and 3.18 to allow derivations of sequents whose right-hand side configurations contain `error` as their syntactic component when a division by zero takes place. For the time being, just follow the same non-modular approach as in the case of SOS (Exercise 42); we are going to discuss a more modular approach to handle errors in MSOS in Section 3.8.*

$x \xrightarrow{\{\text{state}=\sigma, \dots\}} \sigma(x)$	(MSOS-LOOKUP)
$\frac{a_1 \rightarrow a'_1}{a_1 + a_2 \rightarrow a'_1 + a_2}$	(MSOS-ADD-ARG1)
$\frac{a_2 \rightarrow a'_2}{a_1 + a_2 \rightarrow a_1 + a'_2}$	(MSOS-ADD-ARG2)
$i_1 + i_2 \rightarrow i_1 +_{Int} i_2$	(MSOS-ADD)
$\frac{a_1 \rightarrow a'_1}{a_1 / a_2 \rightarrow a'_1 / a_2}$	(MSOS-DIV-ARG1)
$\frac{a_2 \rightarrow a'_2}{a_1 / a_2 \rightarrow a_1 / a'_2}$	(MSOS-DIV-ARG2)
$i_1 / i_2 \rightarrow i_1 /_{Int} i_2 \quad \text{when } i_2 \neq 0$	(MSOS-DIV)
$\frac{a_1 \rightarrow a'_1}{a_1 \leq a_2 \rightarrow a'_1 \leq a_2}$	(MSOS-LEQ-ARG1)
$\frac{a_2 \rightarrow a'_2}{i_1 \leq a_2 \rightarrow i_1 \leq a'_2}$	(MSOS-LEQ-ARG2)
$i_1 \leq i_2 \rightarrow i_1 \leq_{Int} i_2$	(MSOS-LEQ)
$\frac{b \rightarrow b'}{\text{not } b \rightarrow \text{not } b'}$	(MSOS-NOT-ARG)
$\text{not true} \rightarrow \text{false}$	(MSOS-NOT-TRUE)
$\text{not false} \rightarrow \text{true}$	(MSOS-NOT-FALSE)
$\frac{b_1 \rightarrow b'_1}{b_1 \text{ and } b_2 \rightarrow b'_1 \text{ and } b_2}$	(MSOS-AND-ARG1)
$\text{false and } b_2 \rightarrow \text{false}$	(MSOS-AND-FALSE)
$\text{true and } b_2 \rightarrow b_2$	(MSOS-AND-TRUE)

Figure 3.17: MSOS(IMP) — MSOS of IMP Expressions ( $i_1, i_2 \in Int$ ;  $x \in VarId$ ;  $a_1, a'_1, a_2, a'_2 \in AExp$ ;  $b, b', b_1, b'_1, b_2 \in BExp$ ;  $\sigma \in State$ ).

$\frac{a \rightarrow a'}{x := a \rightarrow x := a'}$	(MSOS-ASGN-ARG2)
$x := i \xrightarrow{\{\text{state}=\sigma, \text{state}'=\sigma[i/x], \dots\}} \text{skip}$	(MSOS-ASGN)
$\frac{s_1 \rightarrow s'_1}{s_1 ; s_2 \rightarrow s'_1 ; s_2}$	(MSOS-SEQ-ARG1)
$\text{skip} ; s_2 \rightarrow s_2$	(MSOS-SEQ-SKIP)
$\frac{b \rightarrow b'}{\text{if } b \text{ then } s_1 \text{ else } s_2 \rightarrow \text{if } b' \text{ then } s_1 \text{ else } s_2}$	(MSOS-IF-ARG1)
$\text{if true then } s_1 \text{ else } s_2 \rightarrow s_1$	(MSOS-IF-TRUE)
$\text{if false then } s_1 \text{ else } s_2 \rightarrow s_2$	(MSOS-IF-FALSE)
$\text{while } b \text{ do } s \rightarrow \text{if } b \text{ then } (s ; \text{while } b \text{ do } s) \text{ else skip}$	(MSOS-WHILE)
$\text{vars } xl ; s \xrightarrow{\{\text{state}'=xl \mapsto 0, \dots\}} s$	(MSOS-VARS)

Figure 3.18: MSOS(IMP) — MSOS of IMP Statements ( $i \in \text{Int}$ ;  $x \in \text{VarId}$ ;  $xl \in \mathbf{List}\{\text{VarId}\}$ ;  $a, a' \in AExp$ ;  $b, b' \in BExp$ ;  $s, s_1, s'_1, s_2 \in Stmt$ ;  $\sigma \in State$ ).

Note, however, that MSOS is more syntactic in nature than SOS, in that each of its reduction rules require syntactic terms in both sides of the transition relation. In particular, that means that, unlike in SOS (see Exercise 40), in MSOS one does not have the option to “dissolve” statements from configurations anymore. Instead, one needs to reduce them to **skip** or some similar syntactic constant; if the original language did not have such a constant then one needs to invent one and add it to the original language or calculus syntax.

### 3.4.2 MSOS in Rewriting Logic

Like big-step and small-step SOS, we can also associate a conditional rewrite rule to each MSOS rule and hereby obtain a rewrite logic theory that faithfully (i.e., step-for-step) captures the MSOS definition. There could be different ways to do this. One way to do it is to first desugar the MSOS definition into a step-for-step equivalent small-step SOS definition as discussed above, and then use the faithful embedding of small-step SOS into rewriting logic discussed in Section 3.3.3. The problem with this approach is that the resulting small-step SOS definition, and implicitly the resulting rewrite logic definition, lack the modularity of the original MSOS definition. In other words, if one wants to extend the MSOS definition with rules that would require global changes to its corresponding SOS definition (e.g., ones adding new semantic components into the label/configuration), then one would also need to manually incorporate all those global changes in the resulting rewriting logic definition.

We first show that any MSOS proof system, say MSOS, can be mechanically translated into a rewriting logic theory, say  $\mathcal{R}_{\text{MSOS}}$ , in such a way that two important aspects of the original MSOS definition are preserved: 1) the corresponding derivation relations are step-for-step equivalent, that is,  $\text{MSOS} \vdash C \rightarrow C'$  if and only if  $\mathcal{R}_{\text{MSOS}} \vdash \mathcal{R}_{C \rightarrow C'}$ , where  $\mathcal{R}_{C \rightarrow C'}$  is the corresponding

syntactic translation of the MSOS sequent  $C \rightarrow C'$  into a rewrite logic sequent; and 2)  $\mathcal{R}_{\text{MSOS}}$  is as modular as MSOS. Second, we apply our generic translation technique on the MSOS formal system  $\text{MSOS}(\text{IMP})$  defined in Section 3.4.1 and obtain a rewrite logic semantics of IMP that is step-for-step equivalent to and as modular as  $\text{MSOS}(\text{IMP})$ . The modularity of  $\text{MSOS}(\text{IMP})$  and of  $\mathcal{R}_{\text{MSOS}(\text{IMP})}$  will pay off when we extend IMP in Section 3.8. Finally, we show how  $\mathcal{R}_{\text{MSOS}(\text{IMP})}$  can be seamlessly defined in Maude, yielding another interpreter for IMP (in addition to those corresponding to the big-step and small-step SOS definitions of IMP in Sections 3.2.3 and 3.3.3).

### Computationally and Modularly Faithful Embedding of MSOS into Rewriting Logic

Our embedding of MSOS into rewriting logic is very similar to that of small-step SOS into rewriting logic, with one important exception: the non-syntactic components of the configuration are all grouped into a *record*, which is a multiset of *attributes*, each attribute being a pair associating appropriate semantic information to a *field*. This allows us to use multiset *matching* (or matching modulo associativity, commutativity and identity) in the corresponding rewrite rules to extract the needed semantic information from the record, thus achieving not only a computationally equivalent embedding of MSOS into rewriting logic, but also one with the same degree of modularity as MSOS.

Formally, let us assume an arbitrary MSOS formal proof system. Let *Attribute* be a fresh sort and let *Record* be the sort **Bag**{*Attribute*} (that means that we assume all the infrastructure needed to define records as comma-separated bags, or multisets, of attributes). For each field *Field* holding semantic contents *Contents* that appears unprimed or primed in any of the labels on any of the transitions in any of rules of the MSOS proof system, let us assume an operation “*Field* =  $\_$  : *Contents*  $\rightarrow$  *Attribute*” (the name of this postfix unary operation is “*Field* =  $\_$ ”). Finally, for each syntactic category *Syntax* used in any transition that appears anywhere in the MSOS proof system, let us define a configuration construct “ $\langle \_, \_ \rangle$  : *Syntax*  $\times$  *Record*  $\rightarrow$  *Configuration*”. We are now ready to define our transformation of an MSOS rule into a rewrite logic rule:

1. Translate it into an SOS rule, as discussed right above Section 3.4.1; we could also go directly from MSOS rules to rewrite logic rules, but we would have to repeat most of the steps from MSOS to SOS that were already discussed;
2. Group all the semantic components in the resulting SOS configurations into a corresponding record, where each semantic component translates into a corresponding attribute using a corresponding label;
3. If the same attributes appear multiple times in a rule and their semantic components are not used anywhere in the rule, then replace them by a generic variable of sort *Record*;
4. Finally, use the technique in Section 3.3.3 to transform the resulting SOS-like rules into rewrite rules, tagging the left-hand configurations with the “ $\circ$ ” symbol.

Applying the steps above, the four MSOS rules discussed right above Section 3.4.1 (namely the ones for variable assignment, output, sequential composition of statements and variable lookup translate into the following rewrite logic rules:

$$\begin{aligned}
& \circ \langle X := I, (\text{state} = \sigma, \rho) \rangle \rightarrow \langle \text{skip}, (\text{state} = \sigma[I/X], \rho) \rangle \\
& \circ \langle \text{output}(i), (\text{output} = \omega, \rho) \rangle \rightarrow \langle \text{skip}, (\text{output} = \omega i, \rho) \rangle \\
& \quad \circ \langle S_1 ; S_2, \rho \rangle \rightarrow \langle S'_1 ; S_2, \rho' \rangle \quad \text{if} \quad \circ \langle S_1, \rho \rangle \rightarrow \langle S'_1, \rho' \rangle \\
& \circ \langle X, (\text{state} = \sigma, \rho) \rangle \rightarrow \langle \sigma(X), (\text{state} = \sigma, \rho) \rangle
\end{aligned}$$



We use the same mechanism as for small-step SOS to obtain the reflexive and transitive many-step closure of the MSOS one-step transition relation. This mechanism was discussed in detail in Section 3.3.3; it essentially consists of adding a configuration marker “ $\star$ ” together with a rule “ $\star C \rightarrow \star C'$  if  $\circ C \rightarrow C'$ ” iteratively applying the one-step relation.

**Theorem 5. (*Faithful embedding of MSOS into rewrite logic*)** *For any MSOS definition MSOS, and any appropriate configurations  $C$  and  $C'$ , the following equivalences hold:*

$$\begin{aligned} \text{MSOS} \vdash C \rightarrow C' &\iff \mathcal{R}_{\text{MSOS}} \vdash \circ \overline{C} \rightarrow^1 \overline{C'} &\iff \mathcal{R}_{\text{MSOS}} \vdash \circ \overline{C} \rightarrow \overline{C'} \\ \text{MSOS} \vdash C \rightarrow^\star C' &\iff \mathcal{R}_{\text{MSOS}} \vdash \star \overline{C} \rightarrow \star \overline{C'} \end{aligned}$$

where  $\mathcal{R}_{\text{MSOS}}$  is the rewrite logic semantic definition obtained from MSOS translating each rule in MSOS as above. (Recall from Section 2.7 that  $\rightarrow^1$  is the one-step rewriting relation obtained by dropping the reflexivity and transitivity rules of rewrite logic. Also, as  $C$  and  $C'$  are parameter-free—parameters only appear in rules— $\overline{C}$  and  $\overline{C'}$  are ground terms.)

Like for the previous embeddings of big-step and small-step SOS into rewriting logic, let us elaborate on the apparent differences between MSOS and  $\mathcal{R}_{\text{MSOS}}$  from a user perspective. The most visible difference is the SOS-like style of writing the rules, namely using configurations instead of labels, which also led to the inheritance of the “ $\circ$ ” mechanism from the embedding of SOS into rewriting logic. Therefore, the equivalent rewriting logic definition is slightly more verbose than the original MSOS definition. On the other hand, it has the advantage that it is more direct than the MSOS definition, in that it eliminates all the notational conventions. Indeed, if we strip MSOS out of its notational conventions and go straight to its essence, we find that that essence is precisely its use of multiset matching to modularly access the semantic components of the configuration. MSOS chose to do this on the labels, using specialized conventions for read-only/write-only/read-write components, while our rewriting logic embedding of MSOS does it in the configurations, uniformly for all semantic components. Where precisely this matching takes place is, in our view, less relevant. What is relevant and brings MSOS its modularity is that multiset matching *does* happen. Therefore, similarly to the big-step and small-step SOS representations in rewriting logic, we conclude that the rewrite theory  $\mathcal{R}_{\text{MSOS}}$  *is* MSOS, and *not* an “encoding” of it.

Like for the previous embeddings of big-step and small-step SOS into rewriting logic, unfortunately,  $\mathcal{R}_{\text{MSOS}}$  (and implicitly MSOS) still lacks the main strengths of rewriting logic, namely context-insensitive and parallel application of rewrite rules. Indeed, the rules of  $\mathcal{R}_{\text{MSOS}}$  can only apply at the top, sequentially, so these rewrite theories corresponding to the faithful embedding of MSOS follow a rather poor rewriting logic style. Like for the previous embeddings, this is not surprising though and does not question that quality of our embeddings. All it says is that MSOS was not meant to have the capabilities of rewriting logic w.r.t. context-insensitivity and parallelism; indeed, all MSOS tries to achieve is to address the lack of modularity of SOS and we believe that it succeeded in doing so. Unfortunately, SOS has several other major problems.

## MSOS of IMP in Rewrite Logic

We here discuss the complete MSOS definition of IMP in rewriting logic, obtained by applying the faithful embedding technique discussed above to the MSOS definition of IMP in Section 3.4.1. Figure 3.19 gives an algebraic definition of configurations as well as needed additional record infrastructure; all the sorts, operations and rules in Figure 3.19 were already discussed either above or in Section

**sorts:**  
*Attribute, Record, Configuration, ExtendedConfiguration*

**subsorts and aliases:**  
*Record* = **Bag**{*Attribute*}  
*Configuration* < *ExtendedConfiguration*

**operations:**  
 $\text{state} = \_ : \text{State} \rightarrow \text{Attribute} \quad // \text{ more "fields" can be added by need}$   
 $\langle \_, \_ \rangle : \text{AExp} \times \text{Record} \rightarrow \text{Configuration}$   
 $\langle \_, \_ \rangle : \text{BExp} \times \text{Record} \rightarrow \text{Configuration}$   
 $\langle \_, \_ \rangle : \text{Stmt} \times \text{Record} \rightarrow \text{Configuration}$   
 $\langle \_ \rangle : \text{Pgm} \rightarrow \text{Configuration}$   
 $\circ\_ : \text{Configuration} \rightarrow \text{ExtendedConfiguration} \quad // \text{ reduce one step}$   
 $\star\_ : \text{Configuration} \rightarrow \text{ExtendedConfiguration} \quad // \text{ reduce all steps}$

**rule:**  
 $\star C \rightarrow \star C' \text{ if } \circ C \rightarrow C' \quad // \text{ where } C, C' \text{ are variables of sort } \text{Configuration}$

Figure 3.19: Configurations and infrastructure for the rewrite logic embedding of MSOS(IMP).

3.3.3. Figure 3.20 gives the rules of the rewriting logic theory  $\mathcal{R}_{\text{MSOS(IMP)}}$  that is obtained by applying the procedure above to the MSOS of IMP in Figures 3.17 and 3.18. Like before, we used the rewriting logic convention that variables start with upper-case letters; if they are greek letters, then we use a similar but larger symbol (e.g.,  $\sigma$  instead of  $\sigma$  for variables of sort *State*, or  $\rho$  instead of  $\rho$  for variables of sort *Record*). The following corollary of Theorem 5 establishes the faithfulness of the representation of the MSOS of IMP in rewriting logic:

**Corollary 4.**  $\text{MSOS(IMP)} \vdash C \rightarrow C' \iff \mathcal{R}_{\text{MSOS(IMP)}} \vdash \circ \overline{C} \rightarrow \overline{C'}$ .

Therefore, there is no perceivable computational difference between the IMP-specific proof system MSOS(IMP) and generic rewriting logic deduction using the IMP-specific rewrite rules in  $\mathcal{R}_{\text{MSOS(IMP)}}$ ; the two are faithfully equivalent. Moreover, by the discussion following Theorem 5,  $\mathcal{R}_{\text{MSOS(IMP)}}$  is also as modular as MSOS(IMP). This will be further emphasized in Section 3.8, where we extend IMP with several features, some of which requiring more attributes.

### ☆ Maude Definition of IMP MSOS

Figure 3.21 shows a straightforward Maude representation of the rewrite theory  $\mathcal{R}_{\text{MSOS(IMP)}}$  in Figures 3.20 and 3.19. The Maude module **IMP-SEMANTICS-MSOS** in Figure 3.21 is executable, so Maude, through its rewriting capabilities, yields an MSOS interpreter for IMP the same way it yielded big-step and small-step SOS interpreters in Sections 3.2.3 and 3.3.3, respectively; for example, the Maude rewrite command

```
Maude> rewrite * < sumPgm > .
```

where **sumPgm** is the first program defined in the module **IMP-PROGRAMS** in Figure 3.4, produces a result of the form (the exact statistics are also irrelevant, so they were replaced by “...”):

```
rewrites: 6335 in ... cpu (... real) (... rewrites/second)
result ExtendedConfiguration: * < skip, state = (n |-> 0 , s |-> 5050) >
```

$$\begin{aligned}
& \circ \langle X, (\text{state} = \sigma, \rho) \rangle \rightarrow \langle \sigma(X), (\text{state} = \sigma, \rho) \rangle \\
& \begin{aligned}
& \circ \langle A_1 + A_2, \rho \rangle \rightarrow \langle A'_1 + A_2, \rho' \rangle \quad \text{if } \circ \langle A_1, \rho \rangle \rightarrow \langle A'_1, \rho' \rangle \\
& \circ \langle A_1 + A_2, \rho \rangle \rightarrow \langle A_1 + A'_2, \rho' \rangle \quad \text{if } \circ \langle A_2, \rho \rangle \rightarrow \langle A'_2, \rho' \rangle \\
& \circ \langle I_1 + I_2, \rho \rangle \rightarrow \langle I_1 +_{Int} I_2, \rho \rangle
\end{aligned} \\
& \begin{aligned}
& \circ \langle A_1 / A_2, \rho \rangle \rightarrow \langle A'_1 / A_2, \rho' \rangle \quad \text{if } \circ \langle A_1, \rho \rangle \rightarrow \langle A'_1, \rho' \rangle \\
& \circ \langle A_1 / A_2, \rho \rangle \rightarrow \langle A_1 / A'_2, \rho' \rangle \quad \text{if } \circ \langle A_2, \rho \rangle \rightarrow \langle A'_2, \rho' \rangle \\
& \circ \langle I_1 / I_2, \rho \rangle \rightarrow \langle I_1 /_{Int} I_2, \rho \rangle \quad \text{if } I_2 \neq 0
\end{aligned} \\
& \begin{aligned}
& \circ \langle A_1 \leq A_2, \rho \rangle \rightarrow \langle A'_1 \leq A_2, \rho' \rangle \quad \text{if } \circ \langle A_1, \rho \rangle \rightarrow \langle A'_1, \rho' \rangle \\
& \circ \langle I_1 \leq A_2, \rho \rangle \rightarrow \langle I_1 \leq A'_2, \rho' \rangle \quad \text{if } \circ \langle A_2, \rho \rangle \rightarrow \langle A'_2, \rho' \rangle \\
& \circ \langle I_1 \leq I_2, \rho \rangle \rightarrow \langle I_1 \leq_{Int} I_2, \rho \rangle
\end{aligned} \\
& \begin{aligned}
& \circ \langle \text{not } B, \rho \rangle \rightarrow \langle \text{not } B', \rho' \rangle \quad \text{if } \circ \langle B, \rho \rangle \rightarrow \langle B', \rho' \rangle \\
& \circ \langle \text{not true}, \rho \rangle \rightarrow \langle \text{false}, \rho \rangle \\
& \circ \langle \text{not false}, \rho \rangle \rightarrow \langle \text{true}, \rho \rangle
\end{aligned} \\
& \begin{aligned}
& \circ \langle B_1 \text{ and } B_2, \rho \rangle \rightarrow \langle B'_1 \text{ and } B_2, \rho' \rangle \quad \text{if } \circ \langle B_1, \rho \rangle \rightarrow \langle B'_1, \rho' \rangle \\
& \circ \langle \text{false and } B_2, \rho \rangle \rightarrow \langle \text{false}, \rho \rangle \\
& \circ \langle \text{true and } B_2, \rho \rangle \rightarrow \langle B_2, \rho \rangle
\end{aligned} \\
& \begin{aligned}
& \circ \langle X := A, \rho \rangle \rightarrow \langle X := A', \rho' \rangle \quad \text{if } \circ \langle A, \rho \rangle \rightarrow \langle A', \rho' \rangle \\
& \circ \langle X := I, (\text{state} = \sigma, \rho) \rangle \rightarrow \langle \text{skip}, (\text{state} = \sigma[I/X], \rho) \rangle
\end{aligned} \\
& \begin{aligned}
& \circ \langle S_1 ; S_2, \rho \rangle \rightarrow \langle S'_1 ; S_2, \rho' \rangle \quad \text{if } \circ \langle S_1, \rho \rangle \rightarrow \langle S'_1, \rho' \rangle \\
& \circ \langle \text{skip} ; S_2, \rho \rangle \rightarrow \langle S_2, \rho \rangle
\end{aligned} \\
& \begin{aligned}
& \circ \langle \text{if } B \text{ then } S_1 \text{ else } S_2, \rho \rangle \rightarrow \langle \text{if } B' \text{ then } S_1 \text{ else } S_2, \rho' \rangle \quad \text{if } \circ \langle B, \rho \rangle \rightarrow \langle B', \rho' \rangle \\
& \circ \langle \text{if true then } S_1 \text{ else } S_2, \rho \rangle \rightarrow \langle S_1, \rho \rangle \\
& \circ \langle \text{if false then } S_1 \text{ else } S_2, \rho \rangle \rightarrow \langle S_2, \rho \rangle
\end{aligned} \\
& \circ \langle \text{while } B \text{ do } S, \rho \rangle \rightarrow \langle \text{if } B \text{ then } (S ; \text{while } B \text{ do } S) \text{ else skip}, \rho \rangle \\
& \circ \langle \text{vars } Xl ; S \rangle \rightarrow \langle S, (\text{state} = Xl \mapsto 0) \rangle
\end{aligned}$$

Figure 3.20:  $\mathcal{R}_{\text{MSOS}(\text{IMP})}$ : the complete MSOS of IMP in rewrite logic

```

mod IMP-CONFIGURATIONS-MSOS is including IMP-SYNTAX + STATE .
  sorts Attribute Record Configuration ExtendedConfiguration .
  subsort Attribute < Record .
  subsort Configuration < ExtendedConfiguration .
  op empty : -> Record .
  op _,_ : Record Record -> Record [assoc comm id: empty] .
  op state'=_ : State -> Attribute .
  op <_,_> : AExp Record -> Configuration .
  op <_,_> : BExp Record -> Configuration .
  op <_,_> : Stmt Record -> Configuration .
  op <_> : Pgm -> Configuration .
  op o_ : Configuration -> ExtendedConfiguration . --- one step
  op *_ : Configuration -> ExtendedConfiguration . --- all steps
  var Cfg Cfg' : Configuration .
  crl * Cfg => * Cfg' if o Cfg => Cfg' .
endm

mod IMP-SEMANTICS-MSOS is including IMP-CONFIGURATIONS-MSOS .
  var X : VarId . var R R' : Record . var Sigma Sigma' : State .
  var I I1 I2 : Int . var X1 : List{VarId} . var S S1 S1' S2 : Stmt .
  var A A' A1 A1' A2 A2' : AExp . var B B' B1 B1' B2 B2' : BExp .

  rl o < X, (state = Sigma, R) > => < Sigma(X), (state = Sigma, R) > .
  crl o < A1 + A2, R > => < A1' + A2, R' > if o < A1, R > => < A1', R' > .
  crl o < A1 + A2, R > => < A1 + A2', R' > if o < A2, R > => < A2', R' > .
  rl o < I1 + I2, R > => < I1 + Int I2, R > .
  crl o < A1 / A2, R > => < A1' / A2, R' > if o < A1, R > => < A1', R' > .
  crl o < A1 / A2, R > => < A1 / A2', R' > if o < A2, R > => < A2', R' > .
  crl o < I1 / I2, R > => < I1 / Int I2, R > if I2 /= 0 .
  crl o < A1 <= A2, R > => < A1' <= A2, R' > if o < A1, R > => < A1', R' > .
  crl o < I1 <= A2, R > => < I1 <= A2', R' > if o < A2, R > => < A2', R' > .
  rl o < I1 <= I2, R > => < I1 <= Int I2, R > .

  crl o < not B, R > => < not B', R' > if o < B, R > => < B', R' > .
  rl o < not true, R > => < false, R > .
  rl o < not false, R > => < true, R > .
  crl o < B1 and B2, R > => < B1' and B2, R' > if o < B1, R > => < B1', R' > .
  rl o < false and B2, R > => < false, R > .
  rl o < true and B2, R > => < B2, R > .

  crl o < X := A, R > => < X := A', R' > if o < A, R > => < A', R' > .
  rl o < X := I, (state = Sigma, R) > => < skip, (state = Sigma[I / X], R) > .
  crl o < S1 ; S2, R > => < S1' ; S2, R' > if o < S1, R > => < S1', R' > .
  rl o < skip ; S2, R > => < S2, R > .
  crl o < if B then S1 else S2, R > => < if B' then S1 else S2, R' > if o < B, R > => < B', R' > .
  rl o < if true then S1 else S2, R > => < S1, R > .
  rl o < if false then S1 else S2, R > => < S2, R > .
  rl o < while B do S, R > => < if B then (S ; while B do S) else skip, R > .
  rl o < vars X1 ; S > => < S, (state = X1 |-> 0) > .
endm

```

Figure 3.21: The MSOS of IMP in Maude, including the definition of configurations

Note that the rewrite command above took the same number of rewrite steps as the similar command executed on the small-step SOS of IMP in Maude discussed in Section 3.3.3, namely 6335. This is not unexpected, because matching is not counted as rewrite steps, no matter how complex it is.

Like for the big-step and small-step SOS definitions in Maude, one can also use any of the general-purpose tools provided by Maude on the MSOS definition above. For example, one can exhaustively search for all possible behaviors of a program using the `search` command:

```
Maude> search * < sumPgm > =>! Cfg:Configuration .
```

As expected, only one behavior will be discovered because our IMP language so far is deterministic.

### 3.4.3 Notes

Modular Structural Operational Semantics (MSOS) was introduced in 1999 by Peter Mosses [31] and since then mainly developed by himself and his collaborators (e.g., [32, 33, 34]). In this section we used the implicitly-modular variant of MSOS introduced in [34], which, as acknowledged by the authors of [34], was partly inspired from discussions with us<sup>2</sup>. To be more precise, we used a slightly simplified version of implicitly-modular MSOS here. In MSOS in its full generality, one can also declare some transitions “unobservable”; to keep the presentation simpler, we here omitted all the observability aspects of MSOS.

The idea of our representation of MSOS into rewriting logic adopted in this section is taken over from [43]. At our knowledge, [22] gives the first representation of MSOS into rewriting logic; this representation also led to the development of the Maude MSOS tool [7]. What is different in the representation in [22] from ours is that the former uses two different types of configuration wrappers, one for the left-hand-side of the transitions and one for the right-hand-side; this was already discussed in Section 3.3.4.

---

<sup>2</sup>In fact, drafts of this book that preceded [34] dropped the implicit labels in MSOS rules for notational simplicity; Peter Mosses found that simple idea worthwhile formalizing within MSOS.